



## 342 Chapter 7 Language Translation Principles

```

<translation-unit> →
    <external-declaration>
    | <translation-unit> <external-declaration>

<external-declaration> →
    <function-definition>
    | <declaration>

<function-definition> →
    <type-specifier> <identifier> ( <parameter-list> ) <compound-statement>
    | <identifier> ( <parameter-list> ) <compound-statement>

<declaration> → <type-specifier> <declarator-list> ;

<type-specifier> → void | char | int

<declarator-list> →
    <identifier>
    | <declarator-list> , <identifier>

<parameter-list> →
    ε
    | <parameter-declaration>
    | <parameter-list> , <parameter-declaration>

<parameter-declaration> → <type-specifier> <identifier>

<compound-statement> → { <declaration-list> <statement-list> }

<declaration-list> →
    ε
    | <declaration>
    | <declaration-list> <declaration>

<statement-list> →
    ε
    | <statement>
    | <statement-list> <statement>

<statement> →
    <compound-statement>
    | <expression-statement>
    | <selection-statement>
    | <iteration-statement>

<expression-statement> → <expression> ;

<selection-statement> →
    if ( <expression> ) <statement>
    | if ( <expression> ) <statement> else <statement>

<iteration-statement> →
    while ( <expression> ) <statement>
    | do <statement> while ( <expression> ) ;

<expression> →
    <relational-expression>
    | <identifier> = <expression>

```

**Figure 7.8**

A grammar for a subset of the C++ language.

